

## excel yourself

Autoren: Ralf Sowa, Christian Hapke

Beachten Sie unsere [Hinweise](#) und [Nutzungsbedingungen](#). Vorgestellte Musterlösungen basieren auf MS-Excel® 2003; sie gelten ausschließlich für aufgezeigte Beispieldaten. Bitte melden Sie uns etwaige Fehler in unseren Informationen – Ihr Feedback ist willkommen: [urs.toolbox@urs-beratung.de](mailto:urs.toolbox@urs-beratung.de)

Eine Übersicht zu unseren Excel-Informationen finden Sie hier: [www.urs-beratung.de/toolbox.htm](http://www.urs-beratung.de/toolbox.htm)

### Excel - Funktionen 3

LINKS, RECHTS, TEIL, TEXT, VERKETTEN, WECHSELN, WERT

## LINKS – RECHTS

Mit den Funktionen LINKS und RECHTS wird einer Zelle eine definierte Anzahl Stellen entnommen (*Teile abgeschnitten*).

	A	B	C	D	E	F
1	Artikel Nr	Kennzeichen				
2	1	A102BBcc4	A	=LINKS(B2;1)		
3	1	A102BBcc4	A102	=LINKS(B3;4)		
4	1	A102BBcc4	cc4	=RECHTS(B4;3)		
5	1	A102BBcc4	4	=RECHTS(B5;1)	8	=C5*2
6	1	A102BBcc4	4	=RECHTS(B6;1)*1	8	=C6*2
7						
8	1	1200,75	1	=LINKS(B8;1)	2	=C8*2
9	1	1200,75	,75	=RECHTS(B9;3)	1,5	=C9*2
10						
11			4	=SUMME(C2:C10)		

In den Beispielen C2 bis C6 werden die Funktionen LINKS und RECHTS auf Texte angewendet. Dass sie auch mit Zahlen arbeiten können, zeigen oben die Zellen C8 und C9.

Zu beachten: Das Ergebnis von LINKS und RECHTS ist stets ein Text, auch wenn die Zelle, aus der etwas herausgeschnitten wurde, eine Zahl enthält.

Schauen Sie in Zelle C11: Dort wird die Summe der Zellen C2:C10 gebildet. Das Ergebnis entspricht der einzigen Zahl, die sich in diesem Bereich befindet: jene 4 aus C6. In dieser Zelle C6 wurde der von =RECHTS(B6;1) gelieferte Text „4“ durch Multiplikation mit 1 in eine Zahl gewandelt.

Spalte E zeigt nochmals, dass mit den Ergebnissen von LINKS und RECHTS gerechnet werden kann, obwohl es Texte sind.

Nebenbei: Im Beispiel oben gibt für Spalte C die „Standard“-Formatierung – in diesem Format „Standard“ werden Texte linksbündig und Zahlen rechtsbündig angezeigt.

## TEIL

Mit der Funktion TEIL wird aus einer Zeichenfolge, beginnend ab einem bestimmten Zeichen, eine bestimmte Anzahl Zeichen ausgeschnitten:

	A	B	C	D
1	Artikel Nr	Kennzeichen		
2	1	A102BBcc4	102	=TEIL(B2;2;3)
3	2	A102CCdd1	CC	=TEIL(B3;5;2)
4	3	B102DDee5	5 ee 102	=RECHTS(B4;1)&" "&TEIL(B4;7;2)&" "&TEIL(B4;2;3)

Formel aus C2 *in Worten*: Liefere jenen Teil aus Zelle B2 als Text, der ab dem zweiten Zeichen beginnt und insgesamt drei Zeichen lang ist.

TEIL liefert – wie auch LINKS und RECHTS – stets Text zurück. Soll das Ergebnis eine Zahl sein, muss mit ihm *gerechnet* werden:

	A	B	C	D
1	Artikel Nr	Kennzeichen		
2	1	A102BBcc4	102	=TEIL(B2;2;3)
3			0	=SUMME(C2:C2)
4				
5			102	=1*TEIL(B2;2;3)
6			106	=TEIL(B2;2;3)+RECHTS(B2;1)
7			208	=SUMME(C5:C6)

## VERKETTEN

Mit VERKETTEN werden verschiedene Texte zu einem Text *verkettet*. Auch hier entstehen stets Texte.

	A	B	C	D	E	F
1	Angela	Merkel	Berlin		Angela Merkel, Berlin	=VERKETTEN(A1;" ";B1;" ";C1)
2	Gerhard	Schröder	Hannover		Gerhard Schröder, Hannover	=A2&" "&B2&" "&C2
3						
4	1	2	12		=A4&B4&C4	
5						
6				0	=SUMME(D4:D5)	
7				12	=D4*1	

E1: Hier sehen sie die *Langversion* bzw. die Funktion VERKETTEN in ihrer eigentlichen Schreibweise: verschiedene Zellen und Texte (oben Leerzeichen) werden miteinander verkettet, das Trennzeichen besteht jeweils aus einem Semikolon.

E2: Dies ist die Kurzversion der Funktion: Ein bloßes „&“ genügt für das VERKETTEN.

D4: Auch Zahlen (in A4 und B4) können verkettet werden. Das Ergebnis ist wiederum stets Text. Enthält der Text Ziffern, ist eine Berechnung erforderlich, um daraus eine Zahl zu machen (in D7 die Multiplikation mit 1).

Bei Nutzung der Funktion VERKETTEN (in der *Langversion*) können maximal 30 Texte miteinander *verkettet* werden. Diese Beschränkung kennt die *Kurzversion* allerdings nicht – mit ihr sind auch mehr Verkettungen möglich (bis zur maximalen Zeichenanzahl pro Zelle von 1.024).

## WECHSELN

Mit der Funktion WECHSELN tauschen Sie eine definierte Zeichenfolge gegen eine andere aus.

	A	B	C	D
1	Angela Merkel, Berlin		Angela Merkel, Schwerin	=WECHSELN(A1;"Berlin";"Schwerin")
2	Angela Merkel, Berlin		Gerhard Schröder, Berlin	=WECHSELN(A2;"Angela Merkel";"Gerhard Schröder")
3				
4	Artikel 1, Artikel 2, Artikel 3		1, 2, 3	=WECHSELN(A4;"Artikel";"")
5	Artikel 1, Artikel 2, Artikel 3		Artikel 1, 2, Artikel 3	=WECHSELN(A5;"Artikel";"";2)
6	Artikel 1, Artikel 2, Artikel 3		Artikel 1, 2, 3	=WECHSELN(WECHSELN(A6;"Artikel";"";3);"Artikel";"";2)
7				
8	2900 Oldenburg		26129 Oldenburg	=WECHSELN(A8;"2900";"26129")

C1 / C2: Mit WECHSELN können Sie Angela Merkel umziehen lassen oder Gerhard Schröder *zurückholen*. Für den Fall, dass Sie insbesondere letzteres nicht wünschen, finden Sie weitere Beispiele:

C4 / C5: Sie können Text-Teile eliminieren (z.B. den Textteil „Artikel“ in Zeile 4) oder bestimmen, dass dies nur für das zweite Auftreten gilt (Zeile 5).

C6: Hier wird die Funktion zweimal angewendet, um das zweite und dritte Auftreten des Textteils „Artikel“ entfallen zu lassen.

C8: Als weiteres Beispiel finden Sie eine Postleitzahländerung. (Weil Postleitzahlen auch mit 0 beginnen können, wird sie hier als Text getauscht.)

Mit WECHSELN können auch Zahlen *getauscht* werden:

	A	B	C	D
1	121212		323232	=WECHSELN(A1;1;3)
2	121212		323232	=WECHSELN(A2;1;3)*1
3				
4	121212		321212	=WECHSELN(A4;1;3;1)
5	121212		12aaa212	=WECHSELN(A5;1;"aaa";2)
6				
7	121212		121212	=WECHSELN(A7;1;7;4)
8	121212		121212	=WECHSELN(A8;123;789)
9				
10			323232	=SUMME(C1:C9)

C1 / C2: Hier wird jede 1 in eine 3 *getauscht*. An der Summe in C10 erkenne Sie, dass auch WECHSELN als Ergebnis stets Text liefert. In C2 wird der von WECHSELN gelieferte Text durch Multiplikation mit 1 in eine Zahl gewandelt.

C4 / C5: In C4 wurde definiert, dass ein Wechsel von „1“ nach „3“ nur für das erste Auftreten gelten soll, in C5 wird die zweite „1“ in die Zeichenfolge „aaa“ getauscht.

C7 / C8: Abgesehen davon, dass hier nun Texte und keine Zahlen mehr vorhanden sind, blieben die Zeichen unverändert. Warum? Für C7 gilt der Tausch von „1“ nach „7“ für das vierte Auftreten – es existiert aber keine vierte „1“ in A7. Für C8 ist ein Tausch der Zeichenkette „123“ nach „789“ bestimmt – es existiert keine Zeichenkette „123“ in A8.

Hier noch eine *kleine Spielerei* mit WECHSELN:

	A	B	C	D
1	22		XXII	=RÖMISCH(A1)
2	22		X X II	=WECHSELN(WIEDERHOLEN("I";A2);"IIIIIIII";"X ")

Auch für WECHSELN gilt: Das Ergebnis ist stets Text. Soll das Ergebnis eine Zahl sein, muss mit ihm *gerechnet* werden.

## WERT

Die Funktion WERT wandelt eine als Text vorliegende Ziffernfolge in eine Zahl. Gleiches geschieht – wie oben mehrfach dargestellt – in dem mit der Ziffernfolge gerechnet wird.

	A	B	C	D
1	10		10	=WERT(A1)
2	10		10	=A2*1
3	100		200	=A3+A3
4	10	10	10	=WERT(A4)
5	a10b10		10	=WERT(TEIL(A5;2;2))
6	a10b10		10	=1*TEIL(A6;2;2)
7				
8			250	=SUMME(C1:C7)

In Spalte A steht (abgesehen von Zelle A4) Text: vor den Ziffern wurde ein Hochkomma eingegeben.

C1 / C2: Die Funktion WERT liefert dasselbe Ergebnis, wie die Multiplikation mit 1; auch ist in beiden Fällen das Ergebnis eine Zahl.

C3: Excel wandelt durch jedes Rechnen – hier Addition – als Text vorliegende Ziffern in eine Zahl.

C4: Die Funktion WERT lässt Zahlen unberührt – eine Zahl bleibt eine Zahl.

C5 / C6: Weitere Beispieldaten.

Die Funktion WERT kann somit auf die Ergebnisse der Textfunktionen LINKS, RECHTS, TEIL, VERKETTEN und WECHSELN angewendet werden, um von ihnen gelieferten Text in eine Zahl zu *wandeln*. – Wie mehrfach erwähnt genügt dafür auch jede Berechnung (z. B. Multiplikation mit 1).

Voraussetzung ist (selbstverständlich), dass der Text ausschließlich Ziffern enthält – andernfalls wird ein Fehler gemeldet:

	A	B	C	D
1	Artikel Nr	Kennzeichen		
2	1	A102BBcc4	102BB	=TEIL(B2;2;5)
3				
4			#WERT!	=TEIL(B2;2;5)*1

## TEXT

Es wird nun niemanden überraschen, dass es auch die Funktion TEXT gibt. Und ebenso wird es nicht überraschen, dass sie den Gegenpart zur Funktion WERT darstellt, und sich entsprechend verhält:

	A	B	C	D
1	10	10		=TEXT(A1;"#")
2	10		10	=TEXT(A2;"#")*1
3	100		200	=TEXT(A3;"#")+TEXT(A3;"#")
4	100		200	=TEXT(2*A4;"#")
5				
6			210	=SUMME(C1:C5)

In Spalte A stehen Zahlen.

C1 / C4 / C6: Die Funktion TEXT wandelt die Zahl in Text um.

C2 / C3: Jede *Berechnung* mit Text, der ausschließlich aus Ziffern besteht, wandelt den Text in eine Zahl.

Vorstehendes stellt allerdings **nicht den Nutzen** der Funktion TEXT dar! Und im Gegensatz zur Funktion WERT (die durch Multiplikation mit 1 ersetzbar ist und insofern *überflüssig* erscheint) hat TEXT einiges zu bieten!

Mit der Funktion TEXT stehen viele Varianten der Zahlenformatierung zur Verfügung:

	A	B	C	D
1	0		0	=TEXT(A1;"0")
2	0			=TEXT(A2;"#")
3	10		10	=TEXT(A3;"0")
4	10		10	=TEXT(A4;"#")
5				
6	1000,75		€ 1000,75	=TEXT(A6;"€0,00")
7	1000,75		€ 1.000,75	=TEXT(A7;"€#.##0,00")
8			2.022 EUR	=TEXT(SUMME(A1:A7);"###0 ""EUR"")
9				
10	10750750,75		€ 10.750.751	=TEXT(A10;"€#.##0")
11	10750750,75	in Tausend	€ 10.751	=TEXT(A11;"€#.##0. ")
12	750750,75	in Millionen	€ 0,8	=TEXT(A12;"€0..,0 ")
13	10750750,75	in Millionen	€ 10,75	=TEXT(A13;"€0..,00 ")
14				
15	12,75		13 %	=TEXT(A15/100;"0 %")
16	12,75		12,75 %	=TEXT(A16/100;"0,00 %")
17				
18	12,5		12:30	=TEXT(A18/24;"HH:MM")
19	12,5		12:30:00	=TEXT(A19/24;"HH:MM:SS")
20	72,5		72:30:00	=TEXT(A20/24;"[h]:MM:SS")

... die beispielsweise in Kombination mit der Funktion VERKETTEN benötigt werden.

	A	B	C
1	USt-Satz	16%	
2	Umsatzerlös	12.345,67	
3	Rundung	12.345,00	=ABRUNDEN(B2;0)
4			
5	Umsatzerlös € 12.345,00		= "Umsatzerlös "&TEXT(B3;"€#.##0,00")
6	zzgl. 16 % USt (€ 1.975,20)		= "zzgl. "&TEXT(B1;"0 %")&" USt ("&TEXT(B3*B1;"€#.###,00")&")"
7	zzgl. 16% UST (1975,20 EUR)		=TEXT(B1;"zzgl."&"0% ""UST """)&TEXT(B1*B3;"(0,00 ""EUR"")"

In B6 finden Sie hier eine *einfache* Lösungsvariante, in B7 eine *professionelle* (sie nutzt mehr Möglichkeiten der Formatierung).